// File: contracts/PolarysExchange.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

import {AssetType, Action} from "./lib/Structs.sol";
import "./lib/OrderEncoder.sol";
import "./interfaces/ITransferHelper.sol";
import "./lib/SignatureVerifier.sol";
import "./lib/UpgradeableContract.sol";
import "./interfaces/IPolarysExchange.sol";

contract PolarysExchange is
    IPolarysExchange,
    OrderEncoder,
    ReentrancyGuard,
    UpgradeableContract
{
    using SafeMath for uint256;

    /* Store orders and data */
    mapping(address => mapping(uint256 => bytes32)) private tokenOrder;
    mapping(bytes32 => bool) private orderFilled;

    mapping(bytes32 => address) private highestBidder;
    mapping(bytes32 => uint256) private highestBid;
    mapping(bytes32 => uint256) private auctionExpirationTime;

    /* TransferHelper Address */
    ITransferHelper public TransferHelper;
    address public feeRecipient; //feeAddress

    string public constant EXCHANGE_NAME = "TAGWEB3";
    string public constant VERSION = "1.0.0";

    bool private safe;
    uint8 public feeRate;

    uint256 public ordersFilled;

    constructor() UpgradeableContract(address(this), msg.sender) {}

    modifier isSecure() {
        require(safe, "insecure call");
        _;
        safe = false;
    }

    modifier notZero() {
        require(msg.value != 0);
        _;
    }

    /* Order Events */
    event OrderCreated(
        address indexed seller,
        address paymentToken,
        address collection,
        uint256 indexed tokenId,
        bytes32 orderHash,
        Order order
    );
    event OrderCanceled(bytes32 orderHash);
    event OrderFilled(
        address indexed seller,
        address indexed buyer,
        address collection,
        uint256 indexed tokenId,
        address paymentToken,
        uint256 price,
        bytes32 orderHash
    );

    /* Auction Events */
    event AuctionCreated(
        address indexed seller,
        address paymentToken,
        address collection,
        uint256 indexed tokenId,
        bytes32 orderHash,
        Auction auction
    );
    event NewBidPlaced(
        address indexed seller,
        address indexed bidder,
        address paymentToken,
        uint256 highestBid,
        address collection,
        uint256 indexed tokenId,
        bytes32 orderHash
    );
    event AuctionCanceled(bytes32 orderHash);
    event AuctionCompleted(
        address indexed seller,
        address indexed highestBidder,
        address collection,
        uint256 indexed tokenId,
        address paymentToken,
        uint256 highestBid,
        bytes32 orderHash
    );

    /**
     * @dev Updates the address of the Transfer Helper contract.
     *      Only the contract administrator can perform this action.
     * @param _transferHelper The address of the new Transfer Helper contract.
     */
    function updateTransferHelper(address _transferHelper) public onlyAdmin {
        TransferHelper = ITransferHelper(_transferHelper);
    }

    /**
     * @dev Execute a sell order or cancel an existing order.
     * @param order The order to be executed or canceled.
     * @param _signature Signature for order verification.
     */
    function executeSellOrCancel(
        Order calldata order,
        bytes memory _signature
    ) external nonReentrant {
        // Compute the hash of the order for verification
        bytes32 orderHash;
        if(order.action == Action.SELL){
            orderHash = _hashOrder(order);
        } else if(order.action == Action.CANCEL){
            orderHash = _orderActive(order.collection, order.tokenId);
        }

        // Verify the provided signature against the order hash and the sender's address
        require(
            SignatureVerifier.verifySignature(
                orderHash,
                _signature,
                msg.sender
            ),
            "Polarys Exchange: Invalid signature for order execution"
        );

        // Verify that the caller is the owner of the token and it is approved for transfer
        require(
            _ownerAndApprovalVerify(order.collection, order.tokenId),
            "Polarys Exchange: You are not the owner or token not approved"
        );

        // Set the safe flag to true before executing the operation
        safe = true;

        // Execute the sell or cancel operation based on the order action
        _executeSellOrCancel(order, orderHash);
    }

    /**
     * @dev Verifies if the caller has approval for the token and is its owner.
     * @param collection Address of the token collection (ERC721 contract).
     * @param tokenId ID of the token to be verified.
     * @return A boolean indicating if the caller has ownership and approval.
     */
    function _ownerAndApprovalVerify(
        address collection,
        uint256 tokenId
    ) private view returns (bool) {
        // Check if the caller has approval for the token and is its owner
        return (_verifyTokenApproval(collection, tokenId) &&
            _validateOwnership(collection, tokenId));
    }

    /**
     * @dev Execute a sell or cancel action for an order.
     * @param order The order containing the action and order details.
     * @param orderHash The unique hash representing the order.
     */
    function _executeSellOrCancel(
        Order calldata order,
        bytes32 orderHash
    ) public isSecure {
        if (order.action == Action.SELL) {
            require(
                _orderActive(order.collection, order.tokenId) == 0,
                "Polarys Exchange: token already on sale"
            );
            _storeData(order.collection, order.tokenId, orderHash);

            emit OrderCreated(
                msg.sender,
                order.paymentToken,
                order.collection,
                order.tokenId,
                orderHash,
                order
            );
        } else if (order.action == Action.CANCEL) {
            delete tokenOrder[order.collection][order.tokenId];

            emit OrderCanceled(orderHash);
        }
    }

    /**
     * @dev Execute a buy order.
     * @param order Order details including asset, collection, tokenId, etc.
     * @param orderHash Hash of the order for verification.
     * @param _signature Signature for order execution.
     * @notice Requires valid signature and order parameters.
     * @notice Requires the order to be verified against the given order hash.
     * @notice Marks the execution as secure.
     */
    function executeBuy(
        Order calldata order,
        bytes32 orderHash,
        bytes memory _signature
    ) external payable notZero nonReentrant {
        require(
            SignatureVerifier.verifySignature(
                orderHash,
                _signature,
                msg.sender
            ),
            "Polarys Exchange: Invalid signature for order execution"
        );
        require(
            _verifyOrdersByTokenId(order, orderHash),
            "Polarys Exchange: Invalid order paremeters"
        );

        safe = true;

        _executeBuy(order, orderHash);
    }

    /**
     * @dev Execute a purchase order
     * @param order Order details
     * @param orderHash Hash of the order
     */
    function _executeBuy(
        Order calldata order,
        bytes32 orderHash
    ) public payable isSecure {
        _transferAssets(
            order.asset,
            order.collection,
            order.seller,
            order.buyer,
            order.tokenId,
            order.tokenAmount
        );

        _transferFunds(order.seller, order.price);

        delete tokenOrder[order.collection][order.tokenId];
        orderFilled[orderHash] = true;

        emit OrderFilled(
            order.seller,
            msg.sender,
            order.collection,
            order.tokenId,
            order.paymentToken,
            order.price,
            orderHash
        );

        _addFilledOrders();
    }

    /**
     * @dev Executes a buy order using ERC20 tokens as payment.
     * @param order The buy order details.
     * @param orderHash The hash of the buy order.
     * @param _signature The signature for order execution.
     * @notice Requires a valid payment token, valid signature, and order parameters.
     * @notice Must be called externally and not during a reentrant call.
     */
    function executeBuyWithERC20(
        Order calldata order,
        bytes32 orderHash,
        bytes memory _signature
    ) external nonReentrant {
        require(
            order.paymentToken != address(0),
            "Polarys Exchange: Invalid payment token"
        );

        require(
            SignatureVerifier.verifySignature(
                orderHash,
                _signature,
                msg.sender
            ),
            "Polarys Exchange: Invalid signature for order execution"
        );

        require(
            _verifyOrdersByTokenId(order, orderHash),
            "Polarys Exchange: Invalid order parameters"
        );

        safe = true;

        _executeBuyWithERC20(order, orderHash);
    }

    /**
     * @dev Execute a buy order using ERC20 payment.
     * Transfers the asset from seller to buyer, and the specified ERC20 payment token from buyer to seller.
     * Deletes the order and emits an OrderFilled event.
     * @param order The order to execute.
     * @param orderHash The hash of the order.
     */
    function _executeBuyWithERC20(
        Order calldata order,
        bytes32 orderHash
    ) public isSecure {
        // Transfer the asset (ERC721 or ERC1155 tokens) from seller to buyer
        _transferAssets(
            order.asset,
            order.collection,
            order.seller,
            order.buyer,
            order.tokenId,
            order.tokenAmount
        );

        // Transfer the specified ERC20 payment token from the buyer to the seller
        _transferERC20(
            order.paymentToken,
            msg.sender,
            order.seller,
            order.price
        );

        // Remove the order from the tokenOrder mapping
        delete tokenOrder[order.collection][order.tokenId];

        // Emit an OrderFilled event to indicate successful execution
        emit OrderFilled(
            order.seller,
            msg.sender,
            order.collection,
            order.tokenId,
            order.paymentToken,
            order.price,
            orderHash
        );

        // Increment the count of filled orders
        _addFilledOrders();
    }

    function _ordersFilled(bytes32 orderHash) public view returns (bool) {
        return (orderFilled[orderHash]);
    }

    /**
     * @dev Executes an auction based on the provided auction parameters and signature.
     * @param auction The auction parameters including collection, tokenId, and action.
     * @param _signature The signature to verify the authenticity of the auction parameters.
     * if the signature is invalid or the sender is not the owner or token is not approved.
     */
    function executeAuction(
        Auction calldata auction,
        bytes memory _signature
    ) public nonReentrant {
        // Generate the order hash from the auction parameters
        bytes32 orderHash;
        if(auction.action == Action.CLAIM){
            orderHash = _orderActive(auction.collection, auction.tokenId);
        } else {
            orderHash = _hashAuction(auction);
        }

        // Verify the signature of the order hash
        require(
            SignatureVerifier.verifySignature(
                orderHash,
                _signature,
                msg.sender
            ),
            "Polarys Exchange: Invalid signature for order execution"
        );

        // Verify that the caller is the owner of the token and it is approved for transfer
        require(
            _ownerAndApprovalVerify(auction.collection, auction.tokenId) || msg.sender == _highestBidder(orderHash),
            "Polarys Exchange: You are not the owner or token not approved"
        );

        // Mark the execution as safe
        safe = true;

        // Execute the auction
        _executeAuction(auction, orderHash);
    }

    /**
     * @dev Execute an auction based on the provided auction details and order hash.
     * @param auction Auction details containing the action type, collection, token ID, and more.
     * @param orderHash Hash of the associated order.
     */
    function _executeAuction(
        Auction calldata auction,
        bytes32 orderHash
    ) public isSecure {
        if (auction.action == Action.RESERVED_PRICE) {
            require(
                _orderActive(auction.collection, auction.tokenId) == 0 &&
                    auction.expirationTime == 0,
                "Polarys Exchange: token already on sale"
            );
            _storeData(auction.collection, auction.tokenId, orderHash);

            emit AuctionCreated(
                msg.sender,
                auction.paymentToken,
                auction.collection,
                auction.tokenId,
                orderHash,
                auction
            );
        } else if (auction.action == Action.CLAIM) {
            bytes32 _orderHash = _orderActive(
                auction.collection,
                auction.tokenId
            );
            require(
                _verifyAuction(auction.expirationTime, orderHash, auction.collection, auction.tokenId),
                "Polarys Exchange: Auction has not ended or invalid caller"
            );

            _transferAssets(
                auction.asset,
                auction.collection,
                auction.seller,
                _highestBidder(_orderHash),
                auction.tokenId,
                auction.tokenAmount
            );

            _transferBack(
                auction.paymentToken,
                auction.seller,
                _highestBid(_orderHash)
            );

            delete tokenOrder[auction.collection][auction.tokenId];
            delete highestBidder[_orderHash];
            delete highestBid[_orderHash];
            delete auctionExpirationTime[_orderHash];
            orderFilled[_orderHash] = true;

            emit AuctionCompleted(
                auction.seller,
                _highestBidder(_orderHash),
                auction.collection,
                auction.tokenId,
                auction.paymentToken,
                _highestBid(_orderHash),
                _orderHash
            );

            emit OrderFilled(
                auction.seller,
                _highestBidder(_orderHash),
                auction.collection,
                auction.tokenId,
                auction.paymentToken,
                _highestBid(_orderHash),
                _orderHash
            );

            _addFilledOrders();
        } else {
            bytes32 _orderHash = _orderActive(
                auction.collection,
                auction.tokenId
            );
            require(
                _highestBidder(_orderHash) == address(0) &&
                    _highestBid(_orderHash) == 0,
                "Polarys Exchange: Auction has started"
            );

            delete tokenOrder[auction.collection][auction.tokenId];

            emit AuctionCanceled(orderHash);
        }
    }

    function _verifyAuction(
        uint256 expirationTime,
        bytes32 orderHash,
        address collection, 
        uint256 tokenId
    ) private view returns (bool) {
        return (expirationTime < block.timestamp
        && msg.sender == _highestBidder(orderHash) || 
        _validateOwnership(collection, tokenId));
    }

    /**
     * @dev Execute a bid in an auction.
     * @param auction The auction details.
     * @param orderHash Hash of the auction order.
     * @param _signature Signature for authentication.
     */
    function executeBid(
        Auction calldata auction,
        bytes32 orderHash,
        bytes memory _signature
    ) external payable notZero nonReentrant {
        // Verify the signature of the auction execution.
        require(
            SignatureVerifier.verifySignature(
                orderHash,
                _signature,
                msg.sender
            ),
            "Polarys Exchange: Invalid signature for auction execution"
        );

        // Verify the validity of auction parameters.
        require(
            _verifyAuctionsByTokenId(auction, orderHash),
            "Polarys Exchange: Invalid auction parameters"
        );

        if(_highestBidder(orderHash) != address(0)){
            require(_expirationTime(orderHash) > block.timestamp, "Polarys Exchange: Auction expired");
        }

        // Mark the execution as safe.
        safe = true;

        // Execute the bid in the auction.
        _executeBid(auction, orderHash);
    }

    /**
     * @dev Executes a bid in an auction.
     * @param auction The auction details.
     * @param orderHash The unique hash identifying the auction order.
     */
    function _executeBid(
        Auction calldata auction,
        bytes32 orderHash
    ) public payable isSecure {
        if (
            _highestBidder(orderHash) != address(0) &&
            _highestBid(orderHash) != 0
        ) {
            // Existing bid present
            address previousBidder = _highestBidder(orderHash);
            uint256 previousBid = _highestBid(orderHash);
            uint256 requiredBid = previousBid.mul(2).div(100);
            require(
                msg.value > requiredBid && auction.highestBid > requiredBid,
                "Polarys Exchange: Minimum required 2% more than current highest bid"
            );
            safeSendETH(previousBidder, previousBid);

            // Update bid information
            highestBidder[orderHash] = msg.sender;
            highestBid[orderHash] = msg.value;
        } else {
            // New bid with no existing bids
            auctionExpirationTime[orderHash] = _calculateAuctionEndTime();
            //auctionExpirationTime[orderHash] = block.timestamp + 480; // Test 
            highestBidder[orderHash] = msg.sender;
            highestBid[orderHash] = msg.value;
        }

        // Emit bid event
        emit NewBidPlaced(
            auction.seller,
            msg.sender,
            auction.paymentToken,
            auction.highestBid,
            auction.collection,
            auction.tokenId,
            orderHash
        );
    }

    /**
     * @dev Execute a bid using ERC20 tokens in an auction.
     * @param auction The auction details.
     * @param orderHash The hash of the auction order.
     * @param _signature The signature for order execution.
     * @dev This function allows users to place bids in an auction using ERC20 tokens.
     *      It verifies the auction parameters, the signature, and the availability of ERC20 tokens.
     *      If the bid is valid, the auction is executed.
     */
    function executeBidWithERC20(
        Auction calldata auction,
        bytes32 orderHash,
        bytes memory _signature
    ) external nonReentrant {
        require(
            auction.paymentToken != address(0),
            "Polarys Exchange: Invalid ERC20"
        );
        require(
            SignatureVerifier.verifySignature(
                orderHash,
                _signature,
                msg.sender
            ),
            "Polarys Exchange: Invalid signature for order execution"
        );
        require(
            _verifyAuctionsByTokenId(auction, orderHash),
            "Polarys Exchange: Order canceled or filled"
        );

        if(_highestBidder(orderHash) != address(0)){
            require(_expirationTime(orderHash) > block.timestamp, "Polarys Exchange: Auction expired");
        }

        safe = true;

        _executeBidWithERC20(auction, orderHash);
    }

    /**
     * @dev Executes a bid in an ERC20-based auction.
     * @param auction The details of the auction.
     * @param orderHash The hash of the auction order.
     * Emits a `NewBidPlaced` event.
     */
    function _executeBidWithERC20(
        Auction calldata auction,
        bytes32 orderHash
    ) public isSecure {
        if (
            _highestBidder(orderHash) != address(0) &&
            _highestBid(orderHash) != 0 // New update: check if auction not has started
        ) {
            address previousBidder = _highestBidder(orderHash);
            uint256 previousBid = _highestBid(orderHash);
            uint256 requiredBid = previousBid.mul(2).div(100);
            require(
                auction.highestBid > requiredBid,
                "Polarys Exchange: Minimum required 2% more than current Bid"
            );

            TransferHelper.executeERC20TransferBack(
                auction.paymentToken,
                previousBidder,
                previousBid
            );

            highestBidder[orderHash] = msg.sender;
            highestBid[orderHash] = auction.highestBid;
        } else {
            // New Update: avoid recalculating end time
            TransferHelper.executeERC20Transfer(
                auction.paymentToken,
                msg.sender,
                address(this),
                auction.highestBid
            );
            auctionExpirationTime[orderHash] = _calculateAuctionEndTime();
            highestBidder[orderHash] = msg.sender;
            highestBid[orderHash] = auction.highestBid;
        }

        emit NewBidPlaced(
            auction.seller,
            msg.sender,
            auction.paymentToken,
            auction.highestBid,
            auction.collection,
            auction.tokenId,
            orderHash
        );
    }

    /**
     * @dev Get the address of the highest bidder for a specific order hash.
     * @param orderHash The hash of the order associated with the auction.
     * @return Address of the highest bidder.
     */
    function _highestBidder(bytes32 orderHash) public view returns (address) {
        return highestBidder[orderHash];
    }

    function _storeData(
        address collection,
        uint256 tokenId,
        bytes32 orderHash
    ) private isSecure {
        tokenOrder[collection][tokenId] = orderHash;
    }

    /**
     * @dev Returns the highest bid amount for a specific order hash.
     * @param orderHash The unique identifier of the order.
     * @return The highest bid amount for the given order hash.
     */
    function _highestBid(bytes32 orderHash) public view returns (uint256) {
        return highestBid[orderHash];
    }

    function _transferBack(
        address paymentToken,
        address to,
        uint256 amount
    ) internal {
        uint256 feeAmount = amount.mul(feeRate).div(100);

        if (paymentToken != address(0)) {
            TransferHelper.executeERC20TransferBack(
                paymentToken,
                to,
                amount.sub(feeAmount)
            );
            TransferHelper.executeERC20TransferBack(
                paymentToken,
                feeRecipient,
                feeAmount
            );
        } else {
            safeSendETH(to, amount.sub(feeAmount));
            safeSendETH(feeRecipient, feeAmount);
        }
    }

    /**
     * @dev Get the expiration time for an order.
     * @param orderHash The unique hash of the order.
     * @return The expiration time of the order.
     */
    function _expirationTime(bytes32 orderHash) public view returns (uint256) {
        return auctionExpirationTime[orderHash];
    }

    function _verifyOrdersByTokenId(
        Order calldata order,
        bytes32 orderHash
    ) internal view returns (bool) {
        return (order.seller != address(0) &&
            order.seller != order.buyer &&
            order.expirationTime > block.timestamp &&
            orderHash == _orderActive(order.collection, order.tokenId) &&
            order.expirationTime > block.timestamp &&
            !_validateOwnership(order.collection, order.tokenId)); //New Update
    }

    function _orderActive(
        address collection,
        uint256 tokenId
    ) private view returns (bytes32) {
        return (tokenOrder[collection][tokenId]);
    }

    function _transferFunds(address seller, uint256 amount) public payable {
        uint256 feeAmount = amount.mul(feeRate).div(100);
        payable(seller).transfer(amount.sub(feeAmount));
        //New Update: No transfer if is address zero
        if (feeRecipient != address(0)) {
            payable(feeRecipient).transfer(feeAmount);
        }
    }

    function _transferERC20(
        address paymentToken,
        address from,
        address to,
        uint256 amount
    ) internal {
        uint256 feeAmount = amount.mul(feeRate).div(100);
        TransferHelper.executeERC20Transfer(
            paymentToken,
            from,
            to,
            amount.sub(feeAmount)
        );
        //New Update: Avoid revert transfer to address zero
        if (feeRecipient != address(0)) {
            TransferHelper.executeERC20Transfer(
                paymentToken,
                from,
                feeRecipient,
                feeAmount
            );
        }
    }

    function _transferAssets(
        AssetType asset,
        address collection,
        address seller,
        address buyer,
        uint256 tokenId,
        uint256 tokenAmount
    ) internal {
        if (asset == AssetType.ERC721) {
            TransferHelper.executeERC721Transfer(
                collection,
                seller,
                buyer,
                tokenId
            );
        } else if (asset == AssetType.ERC1155) {
            TransferHelper.executeERC1155Transfer(
                collection,
                seller,
                buyer,
                tokenId,
                tokenAmount
            );
        }
    }

    function _verifyAuctionsByTokenId(
        Auction memory auction,
        bytes32 orderHash
    ) internal view returns (bool) {
        return (
            //Check if the owner not is zero address
            (auction.seller != address(0) &&
            //seller can't place a bid
            auction.seller != auction.highestBidder &&
            //Check if the auction is correct
            orderHash == _orderActive(auction.collection, auction.tokenId) &&
            //check if no change the time expiration
            auction.expirationTime == _expirationTime(orderHash) &&
            //check twice if bidder not is the owner
            !_validateOwnership(auction.collection, auction.tokenId)));
    }

    function safeSendETH(address recipient, uint256 amount) private {
        require(
            address(this).balance >= amount,
            "Polarys Exchange: Insufficient contract balance"
        );

        (bool success, ) = payable(recipient).call{value: amount}("");
        require(success, "Polarys Exchange: Failed to send ETH");
    }

    function _calculateAuctionEndTime() internal view returns (uint256) {
        uint256 currentTime = block.timestamp;
        uint256 endTime = currentTime + 86400; // 86400 seconds in 24 hours
        return endTime;
    }

    function _validateOwnership(
        address collection_,
        uint256 tokenId_
    ) internal view returns (bool) {
        return IERC721(collection_).ownerOf(tokenId_) == msg.sender;
    }

    /**
     * @dev Update the fee rate and fee recipient address.
     * @param newAddr The new address to receive fees.
     * @param newRate The new fee rate as a percentage.
     * Requirements:
     * - Caller must have administrative privileges.
     */
    function updateFeeRateAndFeeRecipient(
        address newAddr,
        uint8 newRate
    ) external onlyAdmin {
        feeRate = newRate;
        feeRecipient = newAddr;
    }

    function _addFilledOrders() private {
        ordersFilled++;
    }

    function _verifyTokenApproval(
        address _collection,
        uint256 _tokenId
    ) private view returns (bool) {
        IERC721 collection = IERC721(_collection);
        return (collection.getApproved(_tokenId) == address(TransferHelper) ||
            collection.isApprovedForAll(msg.sender, address(TransferHelper)));
    }
}

// File: contracts/interfaces/IPolarysExchange.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import {Order, Auction} from "../lib/Structs.sol";

interface IPolarysExchange {
    function executeSellOrCancel(
        Order calldata order,
        bytes calldata _signature
    ) external;

    function executeBuy(
        Order calldata order,
        bytes32 orderHash,
        bytes calldata _signature
    ) external payable;

    function executeBuyWithERC20(
        Order calldata order,
        bytes32 orderHash,
        bytes calldata _signature
    ) external;

    function executeAuction(
        Auction calldata auction,
        bytes calldata _signature
    ) external;

    function executeBid(
        Auction calldata auction,
        bytes32 orderHash,
        bytes calldata _signature
    ) external payable;

    function executeBidWithERC20(
        Auction calldata auction,
        bytes32 orderHash,
        bytes calldata _signature
    ) external;
}

// File: contracts/lib/UpgradeableContract.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

contract UpgradeableContract {
    address private _implementation;
    address private _admin;

    event Upgraded(address indexed implementation);

    modifier onlyAdmin() {
        require(msg.sender == _admin, "Only admin can perform this action");
        _;
    }

    constructor(address implementation_, address admin_) {
        require(implementation_ != address(0), "Invalid implementation address");
        require(admin_ != address(0), "Invalid admin address");

        _implementation = implementation_;
        _admin = admin_;
    }

    function implementation() public view returns (address) {
        return _implementation;
    }

    function admin() public view returns (address) {
        return _admin;
    }

    function upgrade(address newImplementation) public onlyAdmin {
        require(newImplementation != address(0), "Invalid new implementation address");
        require(newImplementation != _implementation, "New implementation address must be different");

        _implementation = newImplementation;

        emit Upgraded(newImplementation);
    }

    function updateAdmin(address newAdmin) public onlyAdmin {
        _admin = newAdmin;
    }

    fallback() external payable {
        address _impl = _implementation;
        assembly {
            // Copy the incoming calldata to memory
            let ptr := mload(0x40)
            calldatacopy(ptr, 0, calldatasize())

            // Delegatecall to the implementation contract
            let result := delegatecall(gas(), _impl, ptr, calldatasize(), 0, 0)

            // Copy the return data to memory
            let size := returndatasize()
            returndatacopy(ptr, 0, size)

            // Check the delegatecall result
            switch result
            case 0 {
                revert(ptr, size)
            }
            default {
                return(ptr, size)
            }
        }
    }

    receive() external payable {
        // Fallback function will be invoked automatically
    }
}
// File: contracts/lib/SignatureVerifier.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/utils/cryptography/ECDSA.sol";

library SignatureVerifier {
    using ECDSA for bytes32;

    function verifySignature(
        bytes32 _messageHash,
        bytes memory _signature,
        address _signer
    ) internal pure returns (bool) {
        bytes32 ethSignedMessageHash = _messageHash.toEthSignedMessageHash();
        address recoveredSigner = ethSignedMessageHash.recover(_signature);
        return recoveredSigner == _signer;
    }
}

// File: contracts/interfaces/ITransferHelper.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface ITransferHelper {
    function addOperator(address _operator) external;

    function removeOperator(address _operator) external;

    function executeERC721Transfer(
        address _tokenAddress,
        address _from,
        address _to,
        uint256 _tokenId
    ) external;

    function executeERC20Transfer(
        address _tokenAddress,
        address _from,
        address _to,
        uint256 _amount
    ) external;

    function executeERC20TransferBack(
        address _tokenAddress,
        address _to,
        uint256 _amount
    ) external;

    function executeERC1155Transfer(
        address _tokenAddress,
        address _from,
        address _to,
        uint256 _tokenId,
        uint256 _amount
    ) external;
}

// File: contracts/lib/OrderEncoder.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import {Order, Auction} from "./Structs.sol";

contract OrderEncoder {
    function _hashOrder(Order calldata order) public pure returns (bytes32) {
        return keccak256(abi.encode(order));
    }

    function _hashKeys(
        address trader,
        address collection,
        uint256 tokenId,
        uint256 expirationTime
    ) internal pure returns (bytes32) {
        return
            keccak256(
                abi.encodePacked(
                    trader,
                    collection,
                    tokenId,
                    expirationTime
                )
            );
    }

    function _hashAuction(
        Auction calldata auction
    ) public pure returns (bytes32) {
        return keccak256(abi.encode(auction));
    }
}

// File: contracts/lib/Structs.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

enum AssetType {
    ERC721,
    ERC1155
}

enum Action {
    SELL,
    BUY,
    RESERVED_PRICE,
    AUCTION_BID,
    CLAIM,
    CANCEL
}

struct Order {
    AssetType asset;
    Action action;
    address collection;
    address paymentToken;
    address seller; 
    address buyer; 
    uint256 tokenId;
    uint256 price;
    uint256 expirationTime; 
    uint256 tokenAmount; //Option for ERC1155
}

struct Auction {
    AssetType asset;
    Action action; 
    address collection; 
    address paymentToken; 
    address seller;
    address highestBidder;
    uint256 tokenId; 
    uint256 reservedPrice;
    uint256 highestBid; 
    uint256 startTime;
    uint256 expirationTime; 
    uint256 tokenAmount; 
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/cryptography/ECDSA.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/cryptography/ECDSA.sol)

pragma solidity ^0.8.0;

import "../Strings.sol";

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 */
library ECDSA {
    enum RecoverError {
        NoError,
        InvalidSignature,
        InvalidSignatureLength,
        InvalidSignatureS,
        InvalidSignatureV // Deprecated in v4.8
    }

    function _throwError(RecoverError error) private pure {
        if (error == RecoverError.NoError) {
            return; // no error: do nothing
        } else if (error == RecoverError.InvalidSignature) {
            revert("ECDSA: invalid signature");
        } else if (error == RecoverError.InvalidSignatureLength) {
            revert("ECDSA: invalid signature length");
        } else if (error == RecoverError.InvalidSignatureS) {
            revert("ECDSA: invalid signature 's' value");
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature` or error string. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     *
     * Documentation for signature generation:
     * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
     * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes memory signature) internal pure returns (address, RecoverError) {
        if (signature.length == 65) {
            bytes32 r;
            bytes32 s;
            uint8 v;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            /// @solidity memory-safe-assembly
            assembly {
                r := mload(add(signature, 0x20))
                s := mload(add(signature, 0x40))
                v := byte(0, mload(add(signature, 0x60)))
            }
            return tryRecover(hash, v, r, s);
        } else {
            return (address(0), RecoverError.InvalidSignatureLength);
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature`. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     */
    function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, signature);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
     *
     * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes32 r, bytes32 vs) internal pure returns (address, RecoverError) {
        bytes32 s = vs & bytes32(0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff);
        uint8 v = uint8((uint256(vs) >> 255) + 27);
        return tryRecover(hash, v, r, s);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
     *
     * _Available since v4.2._
     */
    function recover(bytes32 hash, bytes32 r, bytes32 vs) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, r, vs);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
     * `r` and `s` signature fields separately.
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, uint8 v, bytes32 r, bytes32 s) internal pure returns (address, RecoverError) {
        // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
        // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
        // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
        // signatures from current libraries generate a unique signature with an s-value in the lower half order.
        //
        // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
        // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
        // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
        // these malleable signatures as well.
        if (uint256(s) > 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0) {
            return (address(0), RecoverError.InvalidSignatureS);
        }

        // If the signature is valid (and not malleable), return the signer address
        address signer = ecrecover(hash, v, r, s);
        if (signer == address(0)) {
            return (address(0), RecoverError.InvalidSignature);
        }

        return (signer, RecoverError.NoError);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `v`,
     * `r` and `s` signature fields separately.
     */
    function recover(bytes32 hash, uint8 v, bytes32 r, bytes32 s) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from a `hash`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32 message) {
        // 32 is the length in bytes of hash,
        // enforced by the type signature above
        /// @solidity memory-safe-assembly
        assembly {
            mstore(0x00, "\x19Ethereum Signed Message:\n32")
            mstore(0x1c, hash)
            message := keccak256(0x00, 0x3c)
        }
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from `s`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes memory s) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n", Strings.toString(s.length), s));
    }

    /**
     * @dev Returns an Ethereum Signed Typed Data, created from a
     * `domainSeparator` and a `structHash`. This produces hash corresponding
     * to the one signed with the
     * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
     * JSON-RPC method as part of EIP-712.
     *
     * See {recover}.
     */
    function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32 data) {
        /// @solidity memory-safe-assembly
        assembly {
            let ptr := mload(0x40)
            mstore(ptr, "\x19\x01")
            mstore(add(ptr, 0x02), domainSeparator)
            mstore(add(ptr, 0x22), structHash)
            data := keccak256(ptr, 0x42)
        }
    }

    /**
     * @dev Returns an Ethereum Signed Data with intended validator, created from a
     * `validator` and `data` according to the version 0 of EIP-191.
     *
     * See {recover}.
     */
    function toDataWithIntendedValidatorHash(address validator, bytes memory data) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19\x00", validator, data));
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";
import "./math/SignedMath.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `int256` to its ASCII `string` decimal representation.
     */
    function toString(int256 value) internal pure returns (string memory) {
        return string(abi.encodePacked(value < 0 ? "-" : "", toString(SignedMath.abs(value))));
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }

    /**
     * @dev Returns true if the two strings are equal.
     */
    function equal(string memory a, string memory b) internal pure returns (bool) {
        return keccak256(bytes(a)) == keccak256(bytes(b));
    }
}

// File: @openzeppelin/contracts/utils/math/SignedMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/SignedMath.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard signed math utilities missing in the Solidity language.
 */
library SignedMath {
    /**
     * @dev Returns the largest of two signed numbers.
     */
    function max(int256 a, int256 b) internal pure returns (int256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two signed numbers.
     */
    function min(int256 a, int256 b) internal pure returns (int256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two signed numbers without overflow.
     * The result is rounded towards zero.
     */
    function average(int256 a, int256 b) internal pure returns (int256) {
        // Formula from the book "Hacker's Delight"
        int256 x = (a & b) + ((a ^ b) >> 1);
        return x + (int256(uint256(x) >> 255) & (a ^ b));
    }

    /**
     * @dev Returns the absolute unsigned value of a signed value.
     */
    function abs(int256 n) internal pure returns (uint256) {
        unchecked {
            // must be unchecked in order to support `n = type(int256).min`
            return uint256(n >= 0 ? n : -n);
        }
    }
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                // The surrounding unchecked block does not change this fact.
                // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1, "Math: mulDiv overflow");

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10 ** 64) {
                value /= 10 ** 64;
                result += 64;
            }
            if (value >= 10 ** 32) {
                value /= 10 ** 32;
                result += 32;
            }
            if (value >= 10 ** 16) {
                value /= 10 ** 16;
                result += 16;
            }
            if (value >= 10 ** 8) {
                value /= 10 ** 8;
                result += 8;
            }
            if (value >= 10 ** 4) {
                value /= 10 ** 4;
                result += 4;
            }
            if (value >= 10 ** 2) {
                value /= 10 ** 2;
                result += 2;
            }
            if (value >= 10 ** 1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
        }
    }
}